INGENIERÍA TÉCNICA INDUSTRIAL - Esp. MECÁNICA

FUNDAMENTOS MATEMÁTICOS DE LA INGENIERÍA

Práctica nº 3: Sucesiones y series numéricas

RESUMEN: Abordamos en esta práctica el tratamiento con MAPLE de sucesiones numéricas y series. Para ello, utilizaremos la capacidad de MAPLE para definir sucesiones (incluso con reglas de recurrencia) y calcular límites. Para ello debemos introducir algunas nociones de programación con MAPLE. Finalizaremos con el estudio de series numéricas con dos objetivos fundamentales: estudiar el carácter y suma de algunas series.

COMANDOS MÁS IMPORTANTES:

seq, irem, if..fi, limit, Limit, with(plots), display, sum, Sum

Algunos comandos necesarios

Para poder introducir algunos tipos de sucesiones necesitaremos los comandos siguientes, que también pueden usarse en otros ámbitos.

1. Maple dispone de un lenguaje de programación propio que nos permite, entre otras muchas cosas, definir funciones o sucesiones definidas "a trozos". Para ello se utiliza la sentencia condicional **if** de la siguiente forma:

if *condición* **then** *sentencia1* **else** *sentencia2* **fi**

Si la condición *condición* es cierta, entonces se ejecuta *sentencia1* y si no es cierta se ejecuta *sentencia2*.

Siempre que se comienza con **if** se debe terminar con **fi**.

Por ejemplo, dos bancos diferentes dan intereses diferentes en sus cuentas corrientes. Definimos la función interes de la siguiente manera:

```
> interes:=banco->if banco=1 then 0.04 else 0.03 fi;
> interes(1);
> interes(2);
```

Cuando hay más de una condición, los **if..fi** se pueden encajar unos dentro de otros, siempre teniendo en cuenta que cada **if** abierto hay que cerrarlo con **fi:**

if *condición1* **then** *sentencia1* **else if** *condición2* **then** *sentencia2* **else** *sentencia3* **fi fi**

Por ejemplo, supongamos que en el banco número 2 anterior el interés también varía según la cuenta sea corriente o a plazo fijo. Entonces interes sería:

```
> restart:
> interes:=banco->if banco=1 then 0.04 else if cuenta=1 then
  0.03 else 0.06 fi fi;
> interes(1);
```

```
> cuenta:=1; interes(2);
> cuenta:=2; interes(2);
```

2. Cuando se trabaja con números enteros, es bueno poder distinguir si un entero es par o impar, o si es múltiplo de 4 o no. Un comando que nos da este tipo de información es

irem (*n*, *m*)

Este comando nos da el resto de la división entera del entero n entre m.
Por ejemplo:

```
> irem(25,3);
> irem(334,2);
```

Así irem(n,2)=0 indica que n es par
irem(n,2)=1 indica que n es impar
Ó
irem(n,4)=0 indica que n es múltiplo de 4

Sucesiones numéricas

Definición

Una sucesión de números reales es una aplicación del conjunto de los números naturales N en el cuerpo de los reales R

$$a: N \text{----->} R$$
$$n \text{-------->} a(n)$$

Si conocemos la expresión general de los términos de la sucesión $a(n)$, entonces podemos definir en MAPLE dicha sucesión de forma análoga a como se definen las funciones; así es más fácil el cálculo de sus elementos y su manejo. La forma de hacerlo es mediante el comando

a:= n -> *expresión en n*

La forma de construir una lista de elementos consecutivos de la sucesión *a(n)* se hace mediante la sentencia

seq(a(n), n= inicio . . final)

Ejemplo: Construir, hallar $a(0)$, $a(5)$, $a(1350)$ y los primeros 25 términos de la sucesión con término general

$$a(n) = \frac{1}{n^2}$$

```
> a:=n->1/n^2;
> a(0);
  a(5);
  a(1350);
  seq(a(n), n=1..25);
```

Ejemplo: Obtener los 30 primeros términos de la sucesión $b(n) = \operatorname{sen}\left(\frac{n\pi}{8}\right)$.

```
> b:=n->sin(n*Pi/8);
> seq(b(n),n=1..30);
```

Ejemplo: Obtener los 20 términos de la sucesión $c(n) = \sqrt{4n^2 - 1} - (2n - 1)$.

```
> c:=n->sqrt(4*n^2-1)-(2*n-1);
> seq(c(n), n=1..20);
```

Progresiones aritméticas y geométricas

Una progresión aritmética es una sucesión cuyo término general $\mathrm{a}(n)$ es de la forma

$$\mathrm{a}(n) = b + dn$$

donde b y d son constantes. A d se le llama diferencia de la progresión.

Una progresión geométrica es una sucesión cuyo término general $\mathrm{a}(n)$ es de la forma

$$\mathrm{a}(n) = kr^n$$

donde k y r son constantes. A r se le llama razón de la progresión.

Calcular los primeros 25 términos de las progresiones siguientes: $\mathrm{a}(n) = 1 + 2\,n$, $\mathrm{a}(n) = 4 - 3\,n$, $\mathrm{a}(n) = \left(\frac{1}{2}\right)^n$, $\mathrm{a}(n) = \left(\frac{3}{2}\right)^n$.

```
> a1:=n->1+2*n; a2:=n->4-3*n; a3:=n->(1/2)^n;
  a4:=n->(3/2)^n;
> seq(a1(n),n=1..25);
> seq(a2(n),n=1..25);
> evalf(seq(a3(n),n=1..25));
> evalf(seq(a4(n),n=1..25));
```

Sucesiones definidas a trozos

Hay sucesiones en las que su término general a(n) es diferente si, por ejemplo, n es par o impar o si es múltiplo de 3 o no.

$$a(n) = \begin{cases} \text{expresión 1,} & \text{si n es par} \\ \text{expresión 2,} & \text{si n es impar} \end{cases}$$

En este caso se combinan los comandos **if** e **irem** comentados en la primera sección.

Ejemplo: Definir y calcular algunos términos.

$$\mathbf{a}(n) = \begin{cases} \dfrac{1}{n} & \text{si n es par} \\ \dfrac{n}{n+1} & \text{si n es impar} \end{cases}$$

```
> restart;
  a:=n-> if irem(n,2) = 0 then 1/n else n/(n+1) fi;
> seq(a(n), n=1..10);
> evalf(%);
```

Ejemplo: Definir y calcular algunos términos de la sucesión

$$\mathrm{a}(n) = \begin{cases} n + \dfrac{3}{n} & \text{si n es múltiplo de 3} \\ \sqrt{\dfrac{2\,n}{n+1} + n^2} & \text{si n no es múltipo de 3} \end{cases}$$

```
> restart:
> a:=n->if irem(n,3)=0 th
```


```
+3/n else sqrt(2*n/(n+1))+n^2
```

```
     fi;
> seq(a(n),n=1..10);
> evalf(%);
```

Ejemplo: Definir la siguiente sucesión

$$a(n)=\begin{cases} n+1 & \text{si n es múltiplo de 2} \\ \sqrt{\dfrac{2\,n}{n+1}} & \text{si n no es múltipo de 2} \end{cases}$$

```
> restart:
> a:=n->if irem(n,2)=0 then n+1 else sqrt(2*n/(n+1)) fi;
> seq(a(n),n=1..10);
> evalf(%);
```

Sucesiones recurrentes

Se llama ***sucesión recurrente*** a una sucesión tal que su término general viene dado en función de otros términos anteriores de la sucesión.

Ejemplo: Sea la sucesión dada por $u(1)=\sqrt{2}$, $u(2)=\sqrt{2+\sqrt{2}}$, $u(3)=\sqrt{2+\sqrt{2+\sqrt{2}}}$, ... Calcular los 10 primeros términos de la sucesión.

Claramente la sucesión sigue la regla $u(n)=\sqrt{2+u(n-1)}$, siendo el primer valor u(1)=$\sqrt{2}$. Es importante observar que este tipo de sucesiones necesita que se especifique quién es el primer valor de la sucesión. Como la definición varía si n es 1 o si no lo es, usamos la sentencia **if..fi:**

```
> restart:
> u:=n->if n=1 then sqrt(2) else sqrt(2+u(n-1)) fi;
> seq(u(n),n=1..10);
> evalf(%);
```

Ejemplo: Sea la sucesión dada por $w(1)=24^{\left(\frac{1}{3}\right)}$, $w(2)=\left(24+24^{\left(\frac{1}{3}\right)}\right)^{\left(\frac{1}{3}\right)}$,

$w(3)=\left(24+\left(24+24^{\left(\frac{1}{3}\right)}\right)^{\left(\frac{1}{3}\right)}\right)^{\left(\frac{1}{3}\right)}$, ... Calcular con 20 cifras los primeros 25 términos de esta sucesión.

```
> w:=n->if n=1 then root[3](24) else root[3](24+w(n-1)) fi;
> evalf(seq(w(n),n=1..25),20);
```

Cálculo de límites de sucesiones

El cálculo de límites de sucesiones con MAPLE es una operación directa con el comando

limit(a(n), n=infinity)

Nótese que la orden **Limit** con L mayúscula, nos escribe la notación matemática de límite.
Nota: En principio calcularemos límites sólo de sucesiones que vienen dadas de forma explícita. No trataremos límites de sucesiones recurrentes ni de sucesiones a trozos.

Ejemplo: Calcular el límite de la sucesión $\sqrt{4\,n^2-1}-(2\,n-1)$

Usamos la sentencia **limit:**

```
> restart:
> limit(sqrt(4*n^2-1)-(2*n-1),n=infinity);
```

A veces, para leer mejor los cálculos o comprobar que no hay errores se usa combinado con la sentencia **Limit:**

```
> Limit(sqrt(4*n^2-1)-(2*n-1),
  n=infinity)=limit(sqrt(4*n^2-1)-(2*n-1),n=infinity);
```

Nótese que el símbolo que hay entre el Limit y limit es un =, y no un :=.

Ejemplo: Determinar el tipo de indeterminación y calcular los siguientes límites.

$$\text{a)}\ \lim_{n\to\infty}\left(1-\frac{2}{3+n}\right)^{n} \quad \text{b)}\ \lim_{n\to\infty}\left(\frac{n+1}{n^2}\right)^{\left(\frac{1}{n}\right)} \quad \text{c)}\ \lim_{n\to\infty}\frac{1+n^2}{5^n\,n}$$

```
> Limit((1-2/(3+n))^n,n=infinity)=limit((1-2/(3+n))^n,n=infi
  nity);
> limit(((n+1)/n^2)^(1/n),n=infinity);
> Limit((1+n^2)/(5^n*n),n=infinity); evalf(%);
```

Ejemplo: Calcular los siguientes límites.

$$\text{a)}\ \lim_{n\to\infty}\left(\frac{n^2+11}{n^2+3}\right)^{(n^2)} \qquad \text{b)}\ \lim_{n\to\infty}\frac{\sqrt{n}+1}{(n+1)^3-1}$$

¿Podemos modificar las expresiones de estas sucesiones de forma que se pueda obtener en cada nueva definición un límite con valor 0, ∞ y un valor constante distinto de estos dos?

```
> limit(((n^2+11)/(n^2+3))^(n^2),n=infinity);
> limit((sqrt(n)+1)/((n+1)^3-1),n=infinity);
```

Ejemplo: Calcular el límite de la sucesión $a(n)=\dfrac{\sqrt{n}}{\sqrt{n+\sqrt{n+\sqrt{n}}}}$

```
> Limit(sqrt(n)/(sqrt(n+sqrt(n+sqrt(n)))),n=infinity)=limit(
  sqrt(n)/(sqrt(n+sqrt(n+sqrt(n)))),n=infinity);
```

Representación de sucesiones

Un elemento muy útil en el estudio preliminar del cálculo del límite es la representación gráfica de la sucesión, sobre todo cuando la orden limit falla por algún motivo (con las sucesiones recurrentes, por ejemplo). Para ello debemos seguir el siguiente esquema

1º) Construimos la sucesión finita de puntos del plano con abscisa en valor n y ordenada el valor de la sucesión a(n). Esto lo hacemos con el comando **seq** anteriormente visto

b:=seq([n,a(n)], n=inicio..final)

2º) Representamos la sucesión de puntos en el plano con el comando

plot([b], style=point, symbol=circle)

Ejemplo: Dibujar los primeros 15 términos de la sucesión $a(n) = \sin\left(\frac{n\pi}{8}\right)$.

Primero definimos la sucesión a que es una función definida en los números naturales.

```
> restart:
> a:=n->sin((n*Pi)/8);
> b:=seq([n,a(n)],n=1..15);
> plot([b],style=point, symbol=circle, color=blue);
```

Ejemplo: Dibujar en una misma gráfica los 30 primeros términos de las sucesiones

$a(n) = n$, $b(n) = (-1)^n$, $y\ c(n) = \frac{1}{n}$.

```
> restart:
> a:=n->n; b:=n->(-1)^n; c:=n->1/n;
> A:=seq([n,a(n)],n=1..30); B:=seq([n,b(n)],n=1..30);
  C:=seq([n,c(n)],n=1..30);
> plot({[A],[B],[C]},style=point,color=[blue,red,green],symb
  ol=[box,cross,diamond]);
```

Ejemplo: El número **e**. Estudiemos el comportamiento de la sucesión $a(n) = \left(1 + \frac{1}{n}\right)^n$

```
> restart:
> a:=n->(1+1/n)^n;
  b:=seq([n,evalf(a(n))], n=1..100);
   plot([b], style=point, symbol=circle, color=black);
```

Series numéricas

Definición

Dada una sucesión de números reales $\{a(n)\}$, se define al ***sucesión de sumas parciales*** $\{S(n)\}$ como

$$S(1) = a(1),\ S(2) = a(1) + a(2),\ S(3) = a(1) + a(2) + a(3),...$$

$$S(n) = \sum_{k=1}^{n} a(k)$$

Así se define la ***serie*** de números reales $\{a(n)\}$ como

$$\sum_{n=1}^{\infty} a(n) = \lim_{n \to \infty} S(n)$$

Si ese límite es un número finito, se dice que la serie *converge*. Si no, se llama serie *divergente*.

Cálculo de series en MAPLE

Para calcular la suma de una serie el comando es

sum(`expresión`, n=1..infinity)

La orden **Sum** devuelve la serie pero sin calcular su valor, lo que se puede aprovechar el uso de las dos órdenes para que quede mejor indicado.

Maple sólo sabe calcular la suma de algunos tipos de series. Puede ocurrir que no sepa hallar una suma, o que los cálculos lleven mucho tiempo, o incluso (esto es lo peor) que el programa se quede "colgado". Por esta razón, hay que ser especialmente prudente y guardar el documento antes de introducir la orden sum**.** En este caso, nos queda la vía de estudiar la sucesión de sumas parciales, usando la misma orden **sum** pero con límites finitos:

sum(*expresión***, n=m..k)**

Ejemplo: Calcular la suma de la serie $\sum_{n=1}^{\infty} \frac{1}{n\,(n+1)}$.

En cuanto al valor de la suma, la orden es muy sencilla:

```
> restart;
> sum(1/(n*(n+1)),n=1..infinity);
> Sum(1/(n*(n+1)),n=1..infinity)=sum(1/(n*(n+1)),n=1..infini
  ty);
```

Ejemplo: Estudiemos ahora lo mismo con la serie alternada $\sum_{n=1}^{\infty} \frac{(-1)^n}{1+2\,n^2}$

```
> restart:
> b:=n->(-1)^n/(1+2*n^2);
> sum(b(n), n=1..infinity);
```

En este caso, MAPLE empieza a no ser capaz de darnos una solución satisfatoria. Si intentamos evaluarlo con **evalf**, se nos queda colgado. Para interrumpir el proceso hay que picar el icono STOP de la barra de herramientas:

```
> evalf(%);
```

Para tener una idea de si ha fallado por divergencia o para calcular al menos aproximadamente la suma de esta serie si es convergente es mejor echar mano de las sumas parciales:

```
> S:=n->sum(b(k),k=1..n);
> evalf(seq(S(n),n=1..25));
```

Vemos que las sumas parciales se van acercando a aproximadamente -0.25..
Para tener una mejor información hagamos una gráfica.

```
> g:=seq([n,S(n)],n=1..50):
> plot([g],style=point,symbol=circle);
```

Para asegurarnos más de la aproximación, podemos hacer un dibujo más cercano al límite

```
> plot([g],n=40..50);
```

Y así podemos afirmar que la serie es convergente y que su valor es aproximadamente -0.256.

Ejemplo: Calcular la serie armónica $\sum_{n=1}^{\infty} \frac{1}{n}$. Estudiar gráficamente lo que ocurre.

```
> restart:
> a:=n->1/n;
> sum(a(n),n=1..infinity);
```

Claramente nos indica que diverge a infinito. Vamos a ver gráficamente sus sumas parciales:

```
> S:=n->sum(a(k),k=1..n);
```

```
> g:=seq([n,S(n)],n=1..50):
> plot([g],style=point,symbol=diamond);
```

Así vemos claro cómo va aumentando a medida que n aumenta, o sea, diverge a infinito.

Ejemplo: Calcular la serie armónica generalizada $\sum_{n=1}^{\infty} \frac{1}{n^2}$. Estudiar gráficamente lo que ocurre.

```
> restart:
> a:=n->1/n^2;
> sum(a(n),n=1..infinity);
```

Vemos que nos dice que converge a $\frac{\pi^2}{6}$. Comprobémoslo gráficamente con sus sumas parciales:

```
> S:=n->sum(a(k),k=1..n);
> g:=seq([n,S(n)],n=1..50):
> plot([g],style=point,symbol=diamond);
```

Ejemplo: Calcular las siguientes series a) $\sum_{n=1}^{\infty} \frac{n}{2^n}$, y b) $\sum_{n=1}^{\infty} \left(n^{\left(\frac{1}{n}\right)} - 1\right)^n$

```
> restart:
> a:=n->n/2^n; b:=n->(n^(1/n)-1)^n;
> sum(a(n), n=1..infinity);
> sum(b(n), n=1..infinity);
```

Ya vemos que la segunda serie es más compleja que la primera. MAPLE no puede con ella. Estudiamos sus sumas parciales:

```
> S:=n->sum(b(k),k=1..n);
> g:=seq([n,S(n)],n=1..50):
> evalf(g);
```

Aquí podemos ver claro que una aproximación a la serie es el número 0.2975974915. Además, son cifras exactas pues a partir de n=25 ya no varían.

Ejemplo: Estudiar el carácter de la serie $\sum_{n=1}^{\infty} \frac{1+n^2}{n!}$, o sea, ver si es divergente o convergente.

Primero vemos si MAPLE puede darnos un valor:

```
> restart:
> u:=n->(1+n^2)/n!;
> sum(u(n),n=1..infinity);
```

Claramente es convergente.

Ejemplo: Estudiar el carácter de la serie $\sum_{n=1}^{\infty} \frac{1}{\sqrt{n} - \frac{2}{3}}$

Actuamos igual que antes.

```
> restart:
> w:=n->1/(sqrt(n)-2/3);
> sum(w(n),n=1..infinity);
```

MAPLE no puede calcular esta. Podría ocurrir que divergiera o que convergiera pero no pudiera con ella. Pasamos a las parciales:

```
> S:=n->sum(w(k),k=1..n);
> g:=seq([n,S(n)],n=1..100):
> evalf(g);
```

Vemos que esta va creciendo siempre. Para estar más seguros lo dibujamos:

```
> plot([g],style=point);
```

Es claro que esta serie diverge.